# Maintainer: Sven-Hendrik Haase <svenstaro@gmail.com>
# Contributor: Arkham <arkham at archlinux dot us>
# Contributor: rabyte <rabyte__gmail>

pkgname=bass
pkgver=1.2
pkgrel=13
# Note: Including full name here so people can find it using the long name
pkgdesc="Beneath A Steel Sky is a 2D point-and-click science fiction thriller set in a bleak vision of the future"
arch=('any')
url="http://revolution.co.uk/games/beneath-a-steel-sky/"
license=('custom')
depends=('scummvm>=0.5')
source=(https://downloads.sourceforge.net/scummvm/$pkgname-cd-$pkgver.zip
        $pkgname.sh)
sha512sums=('ff896e723b4dbf2cc52851ee1260c1006aa285b768a727974719d63509bdac1c65be9ea565eb037d5e9e1c8fd8fd76a8c04776bcde46616ca9ccce18ea498d64'
            '89a12d3fa558ddd548bf06a9c3c1fb5dcefcd32256ae9cbaa440dcebf46fb5b3fc3d7b9196b7d9bb0dd85825fccfe0d5f559a94266eaa6b3f7bcb32657aebad8')

package() {
  cd $pkgname-cd-$pkgver

  install -d "$pkgdir"/usr/share/{licenses/,}bass
  install -m 644 readme.txt "$pkgdir"/usr/share/licenses/bass/
  install -m 644 sky.{cpt,dnr,dsk} "$pkgdir"/usr/share/bass/
  install -Dm 755 "$srcdir"/bass.sh "$pkgdir"/usr/bin/bass
}
